

In[1]:=

```
sortCC[polyinds_] := Block[{cent, poly},
  poly = Lookup[indToPtsAssoc, polyinds];
  Lookup[ptsToIndAssoc,
    DeleteDuplicates@
      Flatten[MeshPrimitives[ConvexHullMesh[poly], 1] /. Line -> Sequence, 1]
  ]
];

Clear[sortPointsCC];
sortPointsCC[polyinds_, indTopts_, ptsToInds_] := Block[{cent, ordering, polyPoints},
  polyPoints = Lookup[indTopts, polyinds];
  cent = Mean@polyPoints;
  ordering = Ordering[ArcTan[#[[1]], #[[2]]] &@ (# - cent) & /@ polyPoints];
  Lookup[ptsToInds, Part[polyPoints, ordering]]
]
```

In[4]:=

```
Clear@T1transitionFn;
T1transitionFn[findEdges_, indToPtsAssoc_, vertexToCellG_,
  cellToVertexG_, dSep_ : 0.02] := Block[{edgeind, connectedcellKeys, edge,
  newpts, cellvertIndices, cellvertices, pos, cellpolys, memF, keyscellP,
  selcellKeys, ptToCell, newptsindices, indToPts = indToPtsAssoc, ptsToInds,
  PtIndToCell, keysToMap, cellindicesAssoc, f1, otherkeys, f2,
  polysharingEdge, bag = CreateDataStructure["DynamicArray"],
  vertToCellG = vertexToCellG, cellToVertG = cellToVertexG, testpts},
  If[findEdges != {},
    Scan[
      (
        edgeind = #;
        If[ContainsAll[Keys[indToPts], edgeind],
          (* should be an edge not
            connected to an edge that has already undergone a T1 *)
          connectedcellKeys = DeleteDuplicates[Flatten@Lookup[vertToCellG, edgeind]];
          cellvertIndices = Lookup[cellToVertG, connectedcellKeys];
          edge = Lookup[indToPts, edgeind];
          If[Length[connectedcellKeys] == 1,
            (*edge that is exposed to the void to be merged as a single vertex*)
            newpts = Mean[edge];
            newptsindices = Max[Keys@indToPts] + 1;
            KeyDropFrom[indToPts, edgeind];
            AppendTo[indToPts, newptsindices -> newpts];
            bag["Append", edgeind];
            ptsToInds = AssociationMap[Reverse, indToPts];
            cellToVertG = MapAt[
              DeleteDuplicates[# /. (Alternatives @@ edgeind) -> newptsindices] &,
              cellToVertG, Key[connectedcellKeys /. {z_Integer} -> z]
            ],
            (*else proceed with T1 transition*)
            newpts = With[{midPt = Mean[edge]},
              midPt + dSep Normalize[(# - midPt)] & /@
```

```

        Flatten[RotationTransform[- $\frac{\pi}{2}$ , midPt] /@ {edge}, 1]
    ];
    testpts = With[{midPt = Mean[edge]},
        midPt + 0.000001 Normalize[(# - midPt)] & /@ newpts
    ];
    pos = Position[cellvertIndices, {OrderlessPatternSequence[
        ___, First@edgeind, ___, Last@edgeind, ___]}, {1}];
    polysharingEdge = Extract[cellvertIndices, pos];
    (* the edge should not be part of any  $\Delta$  *)
    If[(AllTrue[polysharingEdge, Length[#]  $\neq$  3] &)&
        ContainsNone[edgeind, Union@* Flatten@* Normal@bag],
        cellvertices = Map[Lookup[indToPts, #] &, cellvertIndices];
        cellpolys = Polygon /@ cellvertices;
        memF = Function[x, RegionMember@x, Listable][Extract[cellpolys, pos]];
        keyscellP = Extract[connectedcellKeys, pos];
        selcellKeys = Thread[keyscellP  $\rightarrow$  memF];
        ptToCell = Quiet[#  $\rightarrow$  First@@Select[selcellKeys, Function[x,
            Last[x][#]]] & /@ testpts /. HoldPattern[_  $\rightarrow$  First[]]  $\rightarrow$  Nothing];
    (* pt to cell *)
    ptToCell = ptToCell /. Thread[testpts  $\rightarrow$  newpts];
    newptsindices = Range[# + 1, # + 2] &[Max[Keys@indToPts]];
    KeyDropFrom[indToPts, edgeind];
    AppendTo[indToPts, Thread[newptsindices  $\rightarrow$  newpts]];
    ptsToInds = AssociationMap[Reverse, indToPts];
    bag["Append", edgeind];
    PtIndToCell = MapAt[ptsToInds, ptToCell, {All, 1}] /. Rule  $\rightarrow$  List;
    (*index to cell*)
    keysToMap = MapAt[Key, PtIndToCell, {All, 2}];
    cellindicesAssoc = AssociationThread[connectedcellKeys, cellvertIndices];

    f1 = Fold[MapAt[Function[x, DeleteDuplicates[x /. Thread[
        edgeind  $\rightarrow$  #2[[1]] ]]], #1, #2[[2]]] &, cellindicesAssoc, keysToMap];
    otherkeys = List@* Key /@ Complement[connectedcellKeys, keyscellP];
    f2 = MapAt[(# /. (Alternatives@@edgeind)  $\rightarrow$  Splice[newptsindices] //
        sortPointsCC[#, indToPts, ptsToInds] &) &, f1, otherkeys];
    AppendTo[cellToVertG, f2];
];
];
vertToCellG = GroupBy[
    Flatten[(Reverse[#, 2] &)@*Thread/@Normal@cellToVertG], First  $\rightarrow$  Last];
]&, findEdges]
];
{indToPts, cellToVertG, vertToCellG}
];

```


## T1 transition

```

In[6]:= hexTile[n_, m_] :=
  With[{hex = Polygon[Table[{Cos[2 Pi k/6] + #, Sin[2 Pi k/6] + #2}, {k, 6}]] &},
    Table[hex[3 i + 3 ((-1)^j + 1)/4, Sqrt[3]/2 j], {i, n}, {j, m}]];
Graphics[{EdgeForm[Black], LightBlue, hexTile[20, 20]}]

```

Out[7]=




```

In[8]:= plt = First[hexTile[1, 4]] // Map[{EdgeForm[Black], LightBlue, #} &, #] & //
  Graphics[#, ImageSize -> Small] &

```

Out[8]=



```

In[9]:= mesh = First[hexTile[1, 4]];

```

```

In[10]:= cells = (MeshPrimitives[#, 0]) & /@ mesh /. Point -> Sequence

```

```

Out[10]= {{{{2., 0.866025}, {2.5, 0.}, {3.5, 0.}, {4., 0.866025}, {3.5, 1.73205}, {2.5, 1.73205}},
  {{3.5, 1.73205}, {4., 0.866025}, {5., 0.866025}, {5.5, 1.73205},
  {5., 2.59808}, {4., 2.59808}}, {{2., 2.59808}, {2.5, 1.73205},
  {3.5, 1.73205}, {4., 2.59808}, {3.5, 3.4641}, {2.5, 3.4641}},
  {{3.5, 3.4641}, {4., 2.59808}, {5., 2.59808}, {5.5, 3.4641}, {5., 4.33013}, {4., 4.33013}}}

```

```

In[11]:= ptsToIndAssoc =
  AssociationThread[# -> Range[Length@#]] &@DeleteDuplicates@Flatten[cells, 1]

```

```

Out[11]= <| {2., 0.866025} -> 1, {2.5, 0.} -> 2, {3.5, 0.} -> 3, {4., 0.866025} -> 4,
  {3.5, 1.73205} -> 5, {2.5, 1.73205} -> 6, {5., 0.866025} -> 7, {5.5, 1.73205} -> 8,
  {5., 2.59808} -> 9, {4., 2.59808} -> 10, {2., 2.59808} -> 11, {3.5, 3.4641} -> 12,
  {2.5, 3.4641} -> 13, {5.5, 3.4641} -> 14, {5., 4.33013} -> 15, {4., 4.33013} -> 16 |>

```

```

In[12]:= indToPtsAssoc = AssociationMap[Reverse, ptsToIndAssoc]

Out[12]= <| 1 → {2., 0.866025}, 2 → {2.5, 0.}, 3 → {3.5, 0.}, 4 → {4., 0.866025},
          5 → {3.5, 1.73205}, 6 → {2.5, 1.73205}, 7 → {5., 0.866025}, 8 → {5.5, 1.73205},
          9 → {5., 2.59808}, 10 → {4., 2.59808}, 11 → {2., 2.59808}, 12 → {3.5, 3.4641},
          13 → {2.5, 3.4641}, 14 → {5.5, 3.4641}, 15 → {5., 4.33013}, 16 → {4., 4.33013} |>

In[13]:= cellToVertex =
  AssociationThread[Range@Length@# → #] &@ (Lookup[ptsToIndAssoc, #] & /@ cells)

Out[13]= <| 1 → {1, 2, 3, 4, 5, 6}, 2 → {5, 4, 7, 8, 9, 10},
          3 → {11, 6, 5, 10, 12, 13}, 4 → {12, 10, 9, 14, 15, 16} |>

In[14]:= vertexToCell =
  GroupBy[Flatten[(Reverse[#, 2] &)@*Thread /@Normal@cellToVertex], First → Last]

Out[14]= <| 1 → {1}, 2 → {1}, 3 → {1}, 4 → {1, 2}, 5 → {1, 2, 3}, 6 → {1, 3}, 7 → {2}, 8 → {2},
          9 → {2, 4}, 10 → {2, 3, 4}, 11 → {3}, 12 → {3, 4}, 13 → {3}, 14 → {4}, 15 → {4}, 16 → {4} |>


```

## Case 1

```

In[15]:= Show[plt,
  Graphics[{Red, PointSize[0.05], Point@indToPtsAssoc[#] & /@ {5, 10}}, ImageSize → Small]

```



```


In[16]:= {$indToPts, $cellToVertex, $vertexToCell} =
  T1transitionFn[{5, 10}, indToPtsAssoc, vertexToCell, cellToVertex, 0.05]

Out[16]= {<| 1 → {2., 0.866025}, 2 → {2.5, 0.}, 3 → {3.5, 0.}, 4 → {4., 0.866025}, 6 → {2.5, 1.73205},
          7 → {5., 0.866025}, 8 → {5.5, 1.73205}, 9 → {5., 2.59808}, 11 → {2., 2.59808},
          12 → {3.5, 3.4641}, 13 → {2.5, 3.4641}, 14 → {5.5, 3.4641}, 15 → {5., 4.33013},
          16 → {4., 4.33013}, 17 → {3.7067, 2.19006}, 18 → {3.7933, 2.14006} |>,
  <| 1 → {1, 2, 3, 4, 18, 17, 6}, 2 → {18, 4, 7, 8, 9},
    3 → {11, 6, 17, 12, 13}, 4 → {17, 18, 9, 14, 15, 16, 12} |>,
  <| 1 → {1}, 2 → {1}, 3 → {1}, 4 → {1, 2}, 18 → {1, 2, 4}, 17 → {1, 3, 4}, 6 → {1, 3}, 7 → {2},
    8 → {2}, 9 → {2, 4}, 11 → {3}, 12 → {3, 4}, 13 → {3}, 14 → {4}, 15 → {4}, 16 → {4} |> }

```

```
In[17]:= Polygon /@ Map[Lookup[$indToPts, #] &, $cellToVertex, {2}] // Values // Graphics[
  {EdgeForm[{Thickness[0.01], Black}], FaceForm[LightBlue], #}, ImageSize → Small] &
```


Out[17]=



## Case 2

```
In[18]:= Show[plt,
  Graphics[{Red, PointSize[0.05], Point@indToPtsAssoc[#] & /@ {1, 2}}, ImageSize → Small]
```

Out[18]=




```
In[19]:= {$indToPts, $cellToVertex, $vertexToCell} =
  T1transitionFn[{{1, 2}}, indToPtsAssoc, vertexToCell, cellToVertex, 0.05]
```

```
Out[19]= {<| 3 → {3.5, 0.}, 4 → {4., 0.866025}, 5 → {3.5, 1.73205}, 6 → {2.5, 1.73205},
  7 → {5., 0.866025}, 8 → {5.5, 1.73205}, 9 → {5., 2.59808}, 10 → {4., 2.59808},
  11 → {2., 2.59808}, 12 → {3.5, 3.4641}, 13 → {2.5, 3.4641}, 14 → {5.5, 3.4641},
  15 → {5., 4.33013}, 16 → {4., 4.33013}, 17 → {2.25, 0.433013} |>, <| 1 → {17, 3, 4, 5, 6},
  2 → {5, 4, 7, 8, 9, 10}, 3 → {11, 6, 5, 10, 12, 13}, 4 → {12, 10, 9, 14, 15, 16} |>,
  <| 17 → {1}, 3 → {1}, 4 → {1, 2}, 5 → {1, 2, 3}, 6 → {1, 3}, 7 → {2}, 8 → {2}, 9 → {2, 4},
  10 → {2, 3, 4}, 11 → {3}, 12 → {3, 4}, 13 → {3}, 14 → {4}, 15 → {4}, 16 → {4} |>}
```

```
In[20]:= Polygon /@ Map[Lookup[$indToPts, #] &, $cellToVertex, {2}] // Values // Graphics[
  {EdgeForm[{Thickness[0.01], Black}], FaceForm[LightBlue], #}, ImageSize → Small] &
```


Out[20]=



### Case 3

```
In[21]:= Show[plt,
  Graphics[{Red, PointSize[0.05], Point@indToPtsAssoc[#] & /@ {1, 6}}, ImageSize → Small]
```

Out[21]=




```
In[22]:= {$indToPts, $cellToVertex, $vertexToCell} =
  T1transitionFn[{{1, 6}}, indToPtsAssoc, vertexToCell, cellToVertex, 0.07]
```

```
Out[22]= {<| 2 → {2.5, 0.}, 3 → {3.5, 0.}, 4 → {4., 0.866025}, 5 → {3.5, 1.73205}, 7 → {5., 0.866025},
  8 → {5.5, 1.73205}, 9 → {5., 2.59808}, 10 → {4., 2.59808}, 11 → {2., 2.59808},
  12 → {3.5, 3.4641}, 13 → {2.5, 3.4641}, 14 → {5.5, 3.4641}, 15 → {5., 4.33013},
  16 → {4., 4.33013}, 17 → {2.18938, 1.33404}, 18 → {2.31062, 1.26404} |>,
  <| 2 → {5, 4, 7, 8, 9, 10}, 4 → {12, 10, 9, 14, 15, 16},
  1 → {18, 2, 3, 4, 5}, 3 → {17, 18, 5, 10, 12, 13, 11} |>,
  <| 5 → {2, 1, 3}, 4 → {2, 1}, 7 → {2}, 8 → {2}, 9 → {2, 4}, 10 → {2, 4, 3}, 12 → {4, 3},
  14 → {4}, 15 → {4}, 16 → {4}, 18 → {1, 3}, 2 → {1}, 3 → {1}, 17 → {3}, 13 → {3}, 11 → {3} |> }
```


```
In[23]:= Polygon /@ Map[Lookup[$indToPts, #] &, $cellToVertex, {2}] // Values // Graphics[
  {EdgeForm[{Thickness[0.01], Black}], FaceForm[LightBlue], #}, ImageSize → Small] &
```

Out[23]=




## Case 4

```
In[24]:= plt2 =
```



```
Show[plt2,
  Graphics[{Red, Thickness[0.02], Line[$indToPts[#] & /@ {17, 18}]}], ImageSize → Small]
```

Out[25]=



```


In[26]:= {{$indToPts, $$cellToVertex, $$vertexToCell} =
  T1transitionFn[{{17, 18}}, $indToPts, $vertexToCell, $cellToVertex, 0.1]
Out[26]= {<| 2 → {2.5, 0.}, 3 → {3.5, 0.}, 4 → {4., 0.866025}, 5 → {3.5, 1.73205},
  7 → {5., 0.866025}, 8 → {5.5, 1.73205}, 9 → {5., 2.59808}, 10 → {4., 2.59808},
  11 → {2., 2.59808}, 12 → {3.5, 3.4641}, 13 → {2.5, 3.4641}, 14 → {5.5, 3.4641},
  15 → {5., 4.33013}, 16 → {4., 4.33013}, 19 → {2.3, 1.38564}, 20 → {2.2, 1.21244} |>,
  <| 2 → {5, 4, 7, 8, 9, 10}, 4 → {12, 10, 9, 14, 15, 16},
  3 → {19, 5, 10, 12, 13, 11}, 1 → {2, 3, 4, 5, 19, 20} |>,
  <| 5 → {2, 3, 1}, 4 → {2, 1}, 7 → {2}, 8 → {2}, 9 → {2, 4}, 10 → {2, 4, 3}, 12 → {4, 3},
  14 → {4}, 15 → {4}, 16 → {4}, 19 → {3, 1}, 13 → {3}, 11 → {3}, 2 → {1}, 3 → {1}, 20 → {1} |> }

```

```

In[27]:= Polygon /@ Map[Lookup[$$indToPts, #] &, $$cellToVertex, {2}] // Values // Graphics[
  {EdgeForm[{Thickness[0.01], Black}], FaceForm[LightBlue], #}, ImageSize → Small] &

```




## Case 5

```

In[28]:= Show[plt,
  Graphics[{Red, PointSize[0.05], Point@indToPtsAssoc[#] & /@ {5, 6}}, ImageSize → Small]

```





```
In[29]:= {$indToPts, $cellToVertex, $vertexToCell} =  
T1transitionFn[{{5, 6}}, indToPtsAssoc, vertexToCell, cellToVertex, 0.07]
```

```
Out[29]:= {<| 1 → {2., 0.866025}, 2 → {2.5, 0.}, 3 → {3.5, 0.}, 4 → {4., 0.866025},  
7 → {5., 0.866025}, 8 → {5.5, 1.73205}, 9 → {5., 2.59808}, 10 → {4., 2.59808},  
11 → {2., 2.59808}, 12 → {3.5, 3.4641}, 13 → {2.5, 3.4641}, 14 → {5.5, 3.4641},  
15 → {5., 4.33013}, 16 → {4., 4.33013}, 17 → {3., 1.66205}, 18 → {3., 1.80205} |>,  
<| 4 → {12, 10, 9, 14, 15, 16}, 1 → {1, 2, 3, 4, 17},  
2 → {17, 4, 7, 8, 9, 10, 18}, 3 → {11, 18, 10, 12, 13} |>,  
<| 12 → {4, 3}, 10 → {4, 2, 3}, 9 → {4, 2}, 14 → {4}, 15 → {4}, 16 → {4}, 1 → {1}, 2 → {1},  
3 → {1}, 4 → {1, 2}, 17 → {1, 2}, 7 → {2}, 8 → {2}, 18 → {2, 3}, 11 → {3}, 13 → {3} |>}
```

```
In[30]:= Polygon /@ Map[Lookup[$indToPts, #] &, $cellToVertex, {2}] // Values // Graphics[  
{EdgeForm[{Thickness[0.01], Black}], FaceForm[LightBlue], #}, ImageSize → Small] &
```

