T1 transition

In[1908]:=
```
sortCC[polyinds_] := Block[{cent, poly},
   poly = Lookup[indToPtsAssoc, polyinds];
   Lookup[ptsToIndAssoc,
    DeleteDuplicates@
     Flatten[MeshPrimitives[ConvexHullMesh[poly], 1] /. Line → Sequence, 1]
   ]
  ];

sortPointCC[polyPoints_] := Block[{cent, ordering},
   cent = Mean@polyPoints;
   ordering = Ordering[ArcTan[#[[1]], #[[2]]] &@(# - cent) & /@ polyPoints];
   Part[polyPoints, ordering]
  ];
```

In[1910]:=
```
hexTile[n_, m_] :=
  With[{hex = Polygon[Table[{Cos[2 Pi k / 6] + #, Sin[2 Pi k / 6] + #2}, {k, 6}]] &},
   Table[hex[3 i + 3 ((-1) ^ j + 1) / 4, Sqrt[3] / 2 j], {i, n}, {j, m}]];
Graphics[{EdgeForm[Black], LightBlue, hexTile[20, 20]}]
```

Out[1911]=



In[1912]:=
```
plt = First[hexTile[1, 4]] // Map[{EdgeForm[Black], LightBlue, #} &, #] & //
  Graphics[#, ImageSize → Small] &
```

Out[1912]=



In[1913]:=
```
mesh = First[hexTile[1, 4]]
```

Out[1913]=
```
{Polygon[ ⊞ ⬢ Number of points: 6 Embedding dimension: 2 ], Polygon[ ⊞ ⬢ Number of points: 6 Embedding dimension: 2 ],
 Polygon[ ⊞ ⬢ Number of points: 6 Embedding dimension: 2 ], Polygon[ ⊞ ⬢ Number of points: 6 Embedding dimension: 2 ]}
```

In[1914]:=
```
pts = Replace[MeshPrimitives[#, 0] & /@ mesh, Point[x : {__Real}] :→ x, {2}];
```

In[1915]:=
```
ptsToIndAssoc =
  AssociationThread[# → Range[Length@#]] &@DeleteDuplicates@Flatten[pts, 1];
```

In[1916]:=
```
indToPtsAssoc = AssociationMap[Reverse, ptsToIndAssoc];
```

In[1917]:=
```
(* let us do T1 about point 5 and 10 *)
```

In[1918]:=
```
Show[plt,
 Graphics[{Red, PointSize[0.05], Point@indToPtsAssoc[#] & /@ {5, 10}}], ImageSize → Small]
```

Out[1918]=



In[1919]:=
```
edge = {indToPtsAssoc[5], indToPtsAssoc[10]};
```

In[1920]:=
```
edgeindex = ptsToIndAssoc /@ edge;
```

In[1921]:=
```
ptindices = Map[Lookup[ptsToIndAssoc, #] &, pts];
```

In[1922]:=
```
cellindices = AssociationThread[Range[4] → ptindices];
```

In[1923]:=
```
keyscells = Keys@cellindices;
```

In[1924]:=
```
pos = Position[Values@Normal@cellindices,
    {OrderlessPatternSequence[___, 5, ___, 10, ___]}, {1}]; // AbsoluteTiming
```

Out[1924]=
```
{0.0000426, Null}
```

In[1925]:=
```
midpoint = Midpoint[edge];
```

In[1926]:=
```
(* take the vertex and rotate anticlockwise *)
```

In[1927]:=
```
dsep = 0.2;
```

In[1928]:=
```
newpts = midpoint + dsep Normalize[(# - midpoint)] & /@
   Flatten[RotationTransform[-π/2, midpoint] /@ {edge}, 1];
```

In[1929]:=
```
Show[plt, Graphics[{Orange, PointSize[0.05], Point@newpts}], ImageSize → Small]
```

Out[1929]=



In[1930]:=
```
memF = Function[x, RegionMember@x, Listable][Extract[mesh, pos]];
```

In[1931]:=
```
pp = Extract[keyscells, pos];
```

In[1932]:=
```
selkeys = Thread[pp → memF]
```

Out[1932]=
```
{2 → RegionMemberFunction[ ⊞ ● Embedding dimension: 2 Region dimension: 2 ],
 3 → RegionMemberFunction[ ⊞ ● Embedding dimension: 2 Region dimension: 2 ]}
```

In[1933]:=
```
xx = # → First @@ Select[selkeys, Function[x, Last[x][#]]] & /@ newpts (*pt to cell *)
```

Out[1933]=
```
{{3.57679, 2.26506} → 3, {3.92321, 2.06506} → 2}
```

In[1934]:=
```
newptsindices = Range[# + 1, # + 2] &[Max[Keys@indToPtsAssoc]]
```

Out[1934]=
```
{17, 18}
```

In[1935]:=
```
AppendTo[indToPtsAssoc, Thread[newptsindices → newpts]];
```

In[1936]:=
```
AppendTo[ptsToIndAssoc, Thread[newpts → newptsindices]];
```

In[1937]:=
```
yy = MapAt[ptsToIndAssoc, xx, {All, 1}] /. Rule → List (*index to cell*)
```

Out[1937]=
```
{{17, 3}, {18, 2}}
```

In[1938]:=
```
keysToMap = MapAt[Key, yy, {All, 2}]
```

Out[1938]=
```
{{17, Key[3]}, {18, Key[2]}}
```

In[1939]:=
```
zz =
 Fold[MapAt[Function[x, DeleteDuplicates[x /. Thread[{5, 10} → #2[[1]]]]], #1, #2[[2]]] &,
  cellindices, keysToMap]
```

Out[1939]=
```
⟨|1 → {1, 2, 3, 4, 5, 6}, 2 → {18, 4, 7, 8, 9},
 3 → {11, 6, 17, 12, 13}, 4 → {12, 10, 9, 14, 15, 16}|⟩
```

In[1940]:=
```
otherkeys = List@*Key /@ Complement[keyscells, pp]
```

Out[1940]=
```
{{Key[1]}, {Key[4]}}
```

In[1941]:=
```
oo =
 MapAt[(# /. (Alternatives @@ edgeindex) → Splice[newptsindices] // sortCC) &, zz, otherkeys]
```

Out[1941]=
```
⟨|1 → {1, 2, 3, 4, 18, 17, 6}, 2 → {18, 4, 7, 8, 9},
 3 → {11, 6, 17, 12, 13}, 4 → {12, 17, 18, 9, 14, 15, 16}|⟩
```

In[1955]:=
```
Graphics[{EdgeForm[{Black}], LightBlue,
  Values[Polygon@Lookup[indToPtsAssoc, #] & /@ oo]}, ImageSize → Small]
```

Out[1955]=



In[1949]:=
```
points = Lookup[indToPtsAssoc, #] & /@ Lookup[oo, {1, 4}];
```

In[1956]:=
```
Polygon@sortPointCC[#] & /@ points //
 Graphics[{EdgeForm[{Thick, Black}], LightBlue, #}, ImageSize → Small] &
```

Out[1956]=